Rules for integrands of the form $(d\,x)^m\,(a+b\,\text{ArcTanh}[c\,x^n])^p$

1. $\int x^m\,(a+b\,\text{ArcTanh}[c\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+$

   1. $\int \frac{(a+b\,\text{ArcTanh}[c\,x^n])^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

      1. $\int \frac{(a+b\,\text{ArcTanh}[c\,x])^p}{x}\,dx$ when $p\in\mathbb{Z}^+$

         1: $\int \frac{a+b\,\text{ArcTanh}[c\,x]}{x}\,dx$

Derivation: Algebraic expansion

Basis: $\text{ArcTanh}[z] == \frac{1}{2}\,\text{Log}[1+z]-\frac{1}{2}\,\text{Log}[1-z]$

Basis: $\text{ArcCoth}[z] == \frac{1}{2}\,\text{Log}\left[1+\frac{1}{z}\right]-\frac{1}{2}\,\text{Log}\left[1-\frac{1}{z}\right]$

Rule:

$$\int \frac{a+b\,\text{ArcTanh}[c\,x]}{x}\,dx \longrightarrow a\int \frac{1}{x}\,dx+\frac{b}{2}\int \frac{\text{Log}[1+c\,x]}{x}\,dx-\frac{b}{2}\int \frac{\text{Log}[1-c\,x]}{x}\,dx$$

$$\longrightarrow a\,\text{Log}[x]-\frac{b}{2}\,\text{PolyLog}[2,-c\,x]+\frac{b}{2}\,\text{PolyLog}[2,c\,x]$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])/x_,x_Symbol] :=
  a*Log[x] - b/2*PolyLog[2,-c*x] + b/2*PolyLog[2,c*x] /;
FreeQ[{a,b,c},x]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])/x_,x_Symbol] :=
  a*Log[x] + b/2*PolyLog[2,-1/(c*x)] - b/2*PolyLog[2,1/(c*x)] /;
FreeQ[{a,b,c},x]
```

2: $\int \frac{(a + b \text{ArcTanh}[c x])^p}{x} \, dx$ when $p - 1 \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{1}{x} == 2 \, \partial_x \text{ArcTanh}\left[1 - \frac{2}{1-c x}\right]$

Rule: If $p - 1 \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \text{ArcTanh}[c x])^p}{x} \, dx \longrightarrow$$

$$2 (a + b \text{ArcTanh}[c x])^p \text{ArcTanh}\left[1 - \frac{2}{1 - c x}\right] - 2 b c p \int \frac{(a + b \text{ArcTanh}[c x])^{p-1} \text{ArcTanh}\left[1 - \frac{2}{1-c x}\right]}{1 - c^2 x^2} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_])^p_/x_,x_Symbol] :=
  2*(a+b*ArcTanh[c*x])^p*ArcTanh[1-2/(1-c*x)] -
  2*b*c*p*Int[(a+b*ArcTanh[c*x])^(p-1)*ArcTanh[1-2/(1-c*x)]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_])^p_/x_,x_Symbol] :=
  2*(a+b*ArcCoth[c*x])^p*ArcCoth[1-2/(1-c*x)] -
  2*b*c*p*Int[(a+b*ArcCoth[c*x])^(p-1)*ArcCoth[1-2/(1-c*x)]/(1-c^2*x^2),x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1]
```

2: $\int \frac{(a + b \operatorname{ArcTanh}[c x^n])^p}{x} \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{F[x^n]}{x} == \frac{1}{n} \operatorname{Subst}\left[\frac{F[x]}{x}, x, x^n\right] \partial_x x^n$

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \frac{(a + b \operatorname{ArcTanh}[c x^n])^p}{x} \, dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int \frac{(a + b \operatorname{ArcTanh}[c x])^p}{x} \, dx, x, x^n\right]$$

Program code:

```
Int[(a_.+b_.*ArcTanh[c_.*x_^n_])^p_./x_,x_Symbol] :=
  1/n*Subst[Int[(a+b*ArcTanh[c*x])^p/x,x],x,x^n] /;
FreeQ[{a,b,c,n},x] && IGtQ[p,0]
```

```
Int[(a_.+b_.*ArcCoth[c_.*x_^n_])^p_./x_,x_Symbol] :=
  1/n*Subst[Int[(a+b*ArcCoth[c*x])^p/x,x],x,x^n] /;
FreeQ[{a,b,c,n},x] && IGtQ[p,0]
```

2: $\int x^m (a+b\,\text{ArcTanh}[c\,x^n])^p\,dx$ when $p\in\mathbb{Z}^+ \land (p==1 \lor n==1 \land m\in\mathbb{Z}) \land m\neq -1$

Derivation: Integration by parts

Basis: $\partial_x (a+b\,\text{ArcTanh}[c\,x^n])^p == b\,c\,n\,p\,\frac{x^{n-1}\,(a+b\,\text{ArcTanh}[c\,x^n])^{p-1}}{1-c^2\,x^{2n}}$

Rule: If $p\in\mathbb{Z}^+ \land (p==1 \lor n==1 \land m\in\mathbb{Z}) \land m\neq -1$, then

$$\int x^m (a+b\,\text{ArcTanh}[c\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}\,(a+b\,\text{ArcTanh}[c\,x^n])^p}{m+1}-\frac{b\,c\,n\,p}{m+1}\int\frac{x^{m+n}\,(a+b\,\text{ArcTanh}[c\,x^n])^{p-1}}{1-c^2\,x^{2n}}\,dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(a+b*ArcTanh[c*x^n])^p/(m+1) -
  b*c*n*p/(m+1)*Int[x^(m+n)*(a+b*ArcTanh[c*x^n])^(p-1)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,m,n},x] && IGtQ[p,0] && (EqQ[p,1] || EqQ[n,1] && IntegerQ[m])  && NeQ[m,-1]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(a+b*ArcCoth[c*x^n])^p/(m+1) -
  b*c*n*p/(m+1)*Int[x^(m+n)*(a+b*ArcCoth[c*x^n])^(p-1)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,m,n},x] && IGtQ[p,0] && (EqQ[p,1] || EqQ[n,1] && IntegerQ[m])  && NeQ[m,-1]
```

3: $\int x^m \left(a + b \operatorname{ArcTanh}\left[c\, x^n\right]\right)^p dx$ when $p - 1 \in \mathbb{Z}^+ \wedge \frac{m+1}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m F[x^n] = \frac{1}{n} \operatorname{Subst}\left[x^{\frac{m+1}{n}-1} F[x],\, x,\, x^n\right] \partial_x x^n$

Rule: If $p - 1 \in \mathbb{Z}^+ \wedge \frac{m+1}{n} \in \mathbb{Z}$, then

$$\int x^m \left(a + b \operatorname{ArcTanh}\left[c\, x^n\right]\right)^p dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a + b \operatorname{ArcTanh}[c\, x]\right)^p dx,\, x,\, x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*ArcTanh[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n},x] && IGtQ[p,1] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*ArcCoth[c*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n},x] && IGtQ[p,1] && IntegerQ[Simplify[(m+1)/n]]
```

4. $\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}$

   1. $\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+$

      1: $\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{ArcTanh}[z] == \frac{\text{Log}[1+z]}{2} - \frac{\text{Log}[1-z]}{2}$

Basis: $\text{ArcCoth}[z] == \frac{\text{Log}[1+z^{-1}]}{2} - \frac{\text{Log}[1-z^{-1}]}{2}$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, then

$$\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx \to \int \text{ExpandIntegrand}\left[x^m \left(a + \frac{b \text{Log}[1 + c x^n]}{2} - \frac{b \text{Log}[1 - c x^n]}{2}\right)^p, x\right] dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[x^m*(a+b*Log[1+c*x^n]/2-b*Log[1-c*x^n]/2)^p,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandIntegrand[x^m*(a+b*Log[1+x^(-n)/c]/2-b*Log[1-x^(-n)/c]/2)^p,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0] && IntegerQ[m]
```

2: $\int x^m \left(a + b \operatorname{ArcTanh}[c\, x^n]\right)^p dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] == k\, \text{Subst}\left[x^{k-1} F\left[x^k\right],\, x,\, x^{1/k}\right] \partial_x x^{1/k}$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^+ \land m \in \mathbb{F}$, let $k \to \text{Denominator}[m]$, then

$$\int x^m \left(a + b \operatorname{ArcTanh}[c\, x^n]\right)^p dx \longrightarrow k\, \text{Subst}\left[\int x^{k\,(m+1)-1} \left(a + b \operatorname{ArcTanh}\left[c\, x^{k\,n}\right]\right)^p dx,\, x,\, x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*ArcTanh[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0] && FractionQ[m]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*ArcCoth[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c},x] && IGtQ[p,1] && IGtQ[n,0] && FractionQ[m]
```

2: $\int x^m \left(a + b \operatorname{ArcTanh}[c\, x^n]\right)^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $\operatorname{ArcTanh}\left[z^{-1}\right] == \operatorname{ArcCoth}[z]$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{Z}^-$, then

$$\int x^m \left(a + b \operatorname{ArcTanh}[c\, x^n]\right)^p \, dx \longrightarrow \int x^m \left(a + b \operatorname{ArcCoth}\left[\frac{x^{-n}}{c}\right]\right)^p \, dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_,x_Symbol] :=
  Int[x^m*(a+b*ArcCoth[x^(-n)/c])^p,x] /;
FreeQ[{a,b,c,m},x] && IGtQ[p,1] && ILtQ[n,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_,x_Symbol] :=
  Int[x^m*(a+b*ArcTanh[x^(-n)/c])^p,x] /;
FreeQ[{a,b,c,m},x] && IGtQ[p,1] && ILtQ[n,0]
```

5: $\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx$ when $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x] = k \, \text{Subst}[x^{k-1} F[x^k], x, x^{1/k}] \, \partial_x x^{1/k}$

Rule: If $p - 1 \in \mathbb{Z}^+ \land n \in \mathbb{F} \land m \in \mathbb{Z}$, let $k \to \text{Denominator}[n]$, then

$$\int x^m (a + b \text{ArcTanh}[c x^n])^p \, dx \longrightarrow k \, \text{Subst}\left[\int x^{k(m+1)-1} (a + b \text{ArcTanh}[c x^{k n}])^p \, dx, x, x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*ArcTanh[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,m},x] && IGtQ[p,1] && FractionQ[n]
```

```
Int[x_^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_])^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*ArcCoth[c*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,m},x] && IGtQ[p,1] && FractionQ[n]
```

2: $\int (d x)^m (a + b \text{ArcTanh}[c x^n]) \, dx$ when $n \in \mathbb{Z} \land m \neq -1$

Derivation: Integration by parts

Basis: If $n \in \mathbb{Z}$, then $\partial_x (a + b \text{ArcTanh}[c x^n]) = \frac{b c n (d x)^{n-1}}{d^{n-1} (1 - c^2 x^{2 n})}$

Rule: If $n \in \mathbb{Z} \land m \neq -1$, then

$$\int (d\,x)^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right) dx \longrightarrow \frac{(d\,x)^{m+1}\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)}{d\,(m+1)} - \frac{b\,c\,n}{d^n\,(m+1)} \int \frac{(d\,x)^{m+n}}{1-c^2\,x^{2\,n}}\,dx$$

–

Program code:

```
Int[(d_*x_)^m_*(a_.+b_.*ArcTanh[c_.*x_^n_.]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcTanh[c*x^n])/(d*(m+1)) -
  b*c*n/(d^n*(m+1))*Int[(d*x)^(m+n)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[n] && NeQ[m,-1]
```

```
Int[(d_*x_)^m_*(a_.+b_.*ArcCoth[c_.*x_^n_.]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCoth[c*x^n])/(d*(m+1)) -
  b*c*n/(d^n*(m+1))*Int[(d*x)^(m+n)/(1-c^2*x^(2*n)),x] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[n] && NeQ[m,-1]
```

3: $\int (d\,x)^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p dx$ when $p \in \mathbb{Z}^+ \land (p == 1 \lor m \in \mathbb{R} \land n \in \mathbb{R})$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(d\,x)^m}{x^m} == 0$

Rule: If $p \in \mathbb{Z}^+ \land (p == 1 \lor m \in \mathbb{F} \land n \in \mathbb{F})$, then

$$\int (d\,x)^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p dx \longrightarrow \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p dx$$

–

Program code:

```
Int[(d_*x_)^m_*(a_.+b_.*ArcTanh[c_.*x_^n_.])^p_.,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*ArcTanh[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p,0] && (EqQ[p,1] || RationalQ[m,n])
```

```
Int[(d_*x_)^m_*(a_.+b_.*ArcCoth[c_.*x_^n_.])^p_.,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*ArcCoth[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p,0] && (EqQ[p,1] || RationalQ[m,n])
```

U: $\int (d\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx$

Rule:

$$\int (d\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx\;\longrightarrow\;\int (d\,x)^m\,\left(a+b\,\text{ArcTanh}\left[c\,x^n\right]\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcTanh[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcTanh[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCoth[c_.*x_^n_.])^p_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcCoth[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```